The Recamán's sequence generates Natural numbers.

Starting from a(0)=0, the n'th term a(n), where n>0, is the previous term minus n i.e a(n) = a(n-1) - n but only if this is both positive and has not been previousely generated.


If the conditions don't hold then a(n) = a(n-1) + n.



Task

 Generate and show here the first 15 members of the sequence.
 Find and show here, the first duplicated number in the sequence.
 Optionally: Find and show here, how many terms of the sequence are needed until all the integers 0..1000, inclusive, are generated.


References

 A005132, The On-Line Encyclopedia of Integer Sequences.
 The Slightly Spooky Recamán Sequence, Numberphile video.
 Recamán's sequence, on Wikipedia.


recaman←{⎕IO←0
    genNext←{
        R←⍵[N-1]-N←≢⍵
        (R<0)∨(R∊⍵):⍵,⍵[N-1]+N
        ⍵,⍵[N-1]-N
    }
    ⎕←'First 15: '
    ⎕←reca←(genNext⍣14),0
    ⎕←'First repetition: '
    ⎕←⊃⌽reca←genNext⍣{⍺≢∪⍺}⊢reca
    ⎕←'Length of sequence containing [0..1000]:'
    ⎕←≢reca←genNext⍣{(⍳1001)∧.∊⊂⍺}⊢reca
}